Commands, Banks and Parameterisation

Table of content

1. Bank file Definition

A bank file is a set of BACON commands.

Each data set can be identified by a label (entry point) and is closed either by **EXIT** or **Return** command.

Examples :

bank1.dat

```
.geometry &
.POIN I 1 X 0 Y 0
      I 2 X 1 Y 0
.DROI I 1 POINT 1 2
.CONT OUVERT LIGNE 1
.DOMA AUTO
.GEN
     MAILLE 1
EXIT

.b_conditions &
.CLM FIX POINT 1
.CLM CHA POINT 2 COMP 1 V 1
EXIT
```

and bank2.dat

```
.POIN I 1 X 0 Y 0
I 2 X 1 Y 0
.DROI I 1 POINTS 1 2
.CONT OUVERT LIGNE 1
.DOMA AUTO
.GENE
MAILLE 1
.CLM FIX POINT 1
.CLM CHA POINT 2 COMP 1 V 1
EXIT
```

Bank file Example

Cylinder with infinite length submitted to internal pressure.
Only a quarter of the structure is modelised.



The bank has to be structured as follows :

- abbreviation definition
- geometry creation
- material definition and their application
- groups creation
- load and boundary condition application
- results storage requirement
- post-processing commands

```
.INIT ! Input address for file pre-processing
.DEL.*
ABRE '/R1NIOB'      '25.E-03'! rayon interne du niobium
ABRE '/R1CUIV'      '26.E-03'! rayon interne du cuivre
ABRE '/R2CUIV'      '29.E-03'! rayon externe du cuivre
ABRE '/Nel_Circ'    '9'      ! Nb elem sur circonference
ABRE '/Nel_Rayniob' '3'      ! Nb elem sur rayon niobium
ABRE '/Nel_Raycui'  '9'      ! Nb elem sur rayon cuivre
ABRE '/Rota'        '90.'    ! Angle de rotation extrusion
ABRE '/Nodint'      '1'      ! Nb Noeuds d'interface
ABRE '/Degre'       '2'      ! Degre du maillage
.3POIN I 1  X  0       0  0
       I 2  X /R1NIOB   0  0
       I 3  X /R1CUIV   0  0
       I 4  X /R2CUIV   0  0
.3DROI  I 1 POINTS 2 3
        I 2 POINTS 3 4
.CONT OUVERT LIGNE 1
      OUVERT LIGNE 2
.DOMA CONT 1
      CONT 2
.GEN DEGRE /Degre
     MODIFIE LIGNE 1   ELEMENT /Nel_Rayniob
     MODIFIE LIGNE 2   ELEMENT /Nel_Raycui
     MAILLE  1 2
.EXT OPERATION 1 RZ /Rota ELEMENT /Nel_Circ INTERF /Nodint COLLER 1
     EXEC 1
.RENUM NOEUDS DEBUT 1;EXEC
       MAILLE DEBUT 1;EXEC
.HYP DEFORMATIONS PLANES
.SEL GROUPE 1 MAILLES NOM "M_TOUT" ; TOUT
     GROUPE 2 NOEUDS NOM "N_FIX_X" ;LIGNES 1 2
     GROUPE 3 NOEUDS NOM "N_FIX_Y" ;BOITE STRUC XI -1. XS .001
     GROUPE 4 MAILLES NOM "M_ARCHIV_STRESS" ; LIGNES 1  2
     GROUPE 5 FACES NOM "F_Pres-Niobium" ;MAILLES ATTRIBUT 1 FACES 4
.MAT I 1 NOM "NIOBIUM"  M 8600. A 7.2E-06 YT 107.E+09 NT 0.39
     I 2 NOM "CUIVRE"   M 9000. A 17.E-06 YT 130.E+09 NT 0.34
.AEL  MAT 1 ATTRIBUT 1
      MAT 2 ATTRIBUT 2
.PHP  GROUPE 1 THICKNESS VALEUR 1.
.CLM FIX GROUPE "N_FIX_X"  COMP 2
     FIX GROUPE « N_FIX_Y"  COMP 1
     PRES GROUPE "F_Pres-Niobium"  V 1.5
.SAI  ARCHIVE GROUPE "M_ARCHIV_STRESS"  STYPE 1310 3310  ! Archivage
RETURN
.VI01 ! Input addres to read file post-processing requirement
.DOC DB "exemple.sdb"
GRAP DIVISE CLOTURE 2 VERTICALES
.DROI I 10 POINTS 1 4
GRAP SELE CLOT 1 ;GRAP CADRE 0;GRAP LEGEN 0
.DES CODE 1411 REFE 4 1 ;VONMISES
GRAP FE 0 DZ 1.25 ;VI
TITRE CXI 45 CYI 80 "Contrainte de Vonmises "
GRAP SELE CLOT 2
SECTION LIGNE 10 FCT 41;VI
.FCT;GRAP LEGE 0 CAXE 1 GL 3 THICK 1 ;AFFICHE NUMERO -1 VALEUR -1
TEXTE ABSCISSES "Rayon [m]" ORDONNEES "Von Mises [N/m2]" FCT 41
     VI 41
TITRE CXI 45 CYI 91 "Evolution radiale de Von Mises  "; EXIT
```

2. Creating a bank file

Three possibilities :

2.1. Open a bank file with your text editor

To edit a file and to write commands in SAMCEF syntax. It is the most advanced way.

2.2. Creation from the content of a Database



Save the content within a bank file named "geom.dat"
`.SAUV BANQUE "geom"`

To load first a database called "model.sdb"
`.DOC DB "model"`

This will save the whole model (command language format) in the *geom.dat* file.

2.3. Creation from an interactive BACON session

Activate the session file.
`MODE MOUCHARD 1`

It will create a <>.spy file.

3. Running a bank file (*.dat)



Read a bank file without entry point
`INPUT "geom.dat"`
With entry point .geometry
`INPUT.geometry "geom.dat"`

- The bank file is executed until an **EXIT** or **RETURN** command is encountered. **RETURN** is used if calling another bank file.
- If no entry point or label address is given, BACON reads and executes commands until an **EXIT** or **RETURN** command is encountered.
- If many labels are written, BACON reads and executes commands between label until next **EXIT** or **RETURN**.

Exercise :
Replace EXIT by RETURN in bank1.dat and create a main bank which calls bank1.dat.

4. Running a spy file (*.spy)

If the bank is a *spy* file, first of all, remove the following lines (with an editor):

```
INPUT INIT
ON
MODE UEC I 0;GRAP RES CLO EFF LOCAL 0 STRU;PICK ACT;TITRE
HEADER « SAMCEF-BAC$ON: &NAME. » ;ABRE 'TXREFRESH' 'VISUALISE';
.SAM NOP 1 -1 NOP5 -1;.MENU$; VISUALISE
```

Then run spy file :



Without entry point
`INPUT "geom.spy" FILTRE`
With an entry point (.start)
`INPUT.start "geom.spy" FILTRE`

5. Guided Exercises

- Guided exercise (N°4) with commands;
- Create a bank file on text editor :
- Create a bank file using .SAUV BANQUE keyboard command (or Export on menu mode)

6. Commands

6.1. General Structure

It exists 2 types of command :

WXYZ : state command with 4 letters. (**GRAP, MODE, ...)**

.XYZ : modelling command + parameters and their values
a point followed by 3 letters.(**.POI, .DROI, .CLM, .MAT, ...)**

The structure is: **`.XYZ PARAMETER <VALUE>`**

Some Parameters :

I	Identificator
X, Y, Z	Coordinates
R, Q, A	Repetition
I, J, K	Step Progression - interpolation
C	Component
VI	Display activation

Example :

```
.3POI i  1 x 0. y 0. z 0.
      i  2 x 1. y 0. z 0.

.DROI  i 1 points 1 2 ; vi

.CLM FIX LIGNE 1 C 1 2 3
```

Notes

- Commands can be written in extenso :

```
.POINTS   .DROITE   .selection          GRAPHISME  MODE
```

or reduced

```
.POI      .DRO      .SEL                GRAP
```

- The command must start at the first column.
- Parameter name must be written only once in a command line.
- To keep a space between parameter and its associated value.
- The format is free and “case insensitive”.

6.2. Main Modelling Commands

Meshing creation		DAO Creation
.NOE	definintion of nodes	**.POI**
.MAI	definition of cells	**.DROITE**
.HYP	definition of the element type	**.CONT**
.SEL	To create a group	**.DOM**
.PHP	Physical property definition	
.MAT	Material definition	
.AEL	Property assignation to cells	
.CLM	Loads and BC's	

Examples :

To create a node number 1 at coordinates 0. 0. 0.

```
.NOE I 1 X 0. Y 0. Z 0.
```

To create a rectangular cell number 1 based on nodes 1 2 3 and 4

```
.MAI I 1 N 1 2 3 4
```

To create a group of elements

```
.SEL GROUP "Shell" MAILLE
I 1 2 3 ..
GROUP "Ground" NOEUD
I 1 2 3 ..
```

To assign a shell behaviour to element number 1

```
.HYP GROUP "Shell" MINDLIN
```

To give a thickness to this shell

```
.PHP GROUP "Shell" THICK VAL 0.001
```

To define a “steel” material

```
.MAT NOM "Steel" YT 2.1e11 NT .3
```

To assign this material to element group

```
.AEL GROUP "Shell" MAT "Steel"
```

To Clamp group “ground” node

```
.CLM GROUP "Ground" FIX
```

6.3. Special Characters

&	All characters before this item are ignored
!	All characters after this item are ignored
/	Start of an abbreviation
;	Many commands in a same line
( )	Arithmetic or mathematical expression
$	Following line (written at the concerned end)

Examples:

```
.POI I 1 X 0. Y 0. ;vi  ! this is a comment
! this is a line of comment
This line also &
```

Remarks:

When parameter first letter is included in the [I-N] range, this parameter is followed by a entire number, the other [A-H] and [O-Z], by a real one.

6.4. State Commands

These commands can be used at any moment. They do not modify the content of the model but only the way to display it.

TITRE	Title of the model
HELP	Documentation
ABRE	Definition of variables (macros)
GRAP	Graphics command
EVALUE	calculator
EDIT	file edition
MODE	operating mode
SYSTEM	system access
Assign	File Assignation

Example:

GRAP VIS 1. 1. 1. ! Definition of a point of view

6.5. I/O commands

6.5.1. General

.FIN	Exporting data for computation (.sdb and .sam files creation)
.SAUVE	Save data in a choosen file name and type.
INPUT	Read and execute a bank file
RETURN	End of bank file execution
.DOC DB	Loading formatted files.

Example:

```
! save of the model in the filename.sdb file.
.SAUVE DB "filename"
```

The **DB** parameter allows to save not only model data but also working space. Resulting file is a **.SDB** file.

6.6. Parameterization

The parameterization (or definition of macro commands) can always be used for any purpose. It only consists on replacing a macro command by its content.

```
ABRE '/my_name' '<chain of characters>'
```

Example :

```
ABRE  '/length'  '10.'
.3POIN I 1  X 0.        Y 0. Z 0.
  I 2  X /length   Y 0. Z 0.
```

The list of all defined macros can be accessed typing **ABRE** command.

6.7. Logical test

Logical tests (FORTRAN syntax) can be used (GT,GE, LT, LE, EQ, …):

Example :

```
ABRE '/VAL'  '10'
#IF (/VAL GT .005)  THEN
.POIN I 1 X /VAL Y 21
#ELSE
.POIN I 1 X 0 Y 20
#ENDIF
```